



**Studio sull'integrazione di basi cartografiche
numeriche diverse per provenienza e scala in
un'unica architettura di sistema informativo
territoriale**

Manuale Operativo



Premessa.....	3
SnapCoverage	4
SnapEnvironment.....	5
Procedura di autocertificazione di un dato.....	6
Controllo.aml	7
<i>Scrivi.aml</i>	13
<i>Esempio di applicazione della procedura di Autocertificazione</i>	14
Software di misura e certificazione della Congruenza esterna:	16
<i>Cong1.aml</i>	16
<i>Cong2.aml</i>	18
Operazioni da eseguire con ArcInfo per rendere congruenti ad una base i dati già acquisiti o da acquisire.	20
<i>Reshape</i>	20
<i>Reshape.aml</i>	22
<i>Barra.menu</i>	23
<i>Barra.aml</i>	24
<i>Set.aml</i>	25
<i>Ap_mospar.aml</i>	26
<i>Barra1.menu</i>	27
<i>Att2.aml</i>	28
Altre Procedure	30
<i>Crea_schiacciata.aml</i>	30
<i>Fpg_in_Sc_line.aml</i>	32
Esempio 1:.....	34
Esempio 2:.....	40
Esempio 3:.....	45



Premessa

Questo è un manuale che contiene procedure e modalità operative in ambiente Arc/Info 7 e ArcView 3.

Il lettore di questo manuale deve preventivamente aver letto la “Relazione Sintetica” nella quale si affronta il problema della congruenza dei dati geografici e fornisce delle metodologie operative per la sua soluzione.

Si presume inoltre che il lettore di questo manuale abbia già una buona conoscenza dell'ambiente Arc/Info e/o dell'ambiente ArcView.



SnapCoverage

Ambiente operativo: ArcEdit di Arc/Info 7

Strati Informativi: aa_ctr, ur_ctr

Scopo: agganciare gli archi di aa_ctr agli archi di ur_ctr

Comandi:

Edit aa_ctr	/*viene editato lo strato aa_ctr che deve essere agganciato allo strato ur_ctr
Drawenvironment arc	/*disegna gli archi dello strato aa_ctr
Snapcoverage ur_ctr	/*specifica la coverage a cui agganciare la coverage in editing
Backcoverage ur_ctr 2	/*attribuisce un colore diverso alla coverage ur_ctr
Backenvironment arc	/*disegna gli archi della coverage ur_ctr come sfondo
Edifeatures arc	/*specifica gli elementi da editare di aa_ctr
snapfeatures arc arc	/*definisce lo snap tra gli archi
Snapping closest 5	/*definisce la distanza entro cui deve essere effettuato lo snap
Sel all	/*seleziona tutti gli archi di aa_ctr
Snap	/*effettua lo snap



SnapEnvironment

Ambiente operativo: ArcEdit di Arc/Info 7
Strati Informativi: aa_ctr, ur_ctr

Scopo: agganciare gli archi di aa_ctr agli archi di ur_ctr, ai_ctr e ag_ctr

Comandi:

Edit ur_ctr	/*viene editato lo strato ur_ctr che deve essere
agganciato allo strato ur_ctr, ai_ctr e ag_ctr	
Drawenvironment arc	/*disegna gli archi dello strato ur_ctr
Edifeatures arc	/*specifica gli elementi da editare di ur_ctr
Snapenvironment ur_ctr arc aa_ctr arc add 5	/*definisce il primo tipo di elemento che deve
agganciare e la prima coverage a cui deve agganciare	
Snapenvironment ur_ctr arc ai_ctr arc add 5	/*definisce il secondo tipo di elemento che deve
agganciare e la seconda coverage a cui deve agganciare	
Snapenvironment ur_ctr arc ag_ctr arc add 5	/*definisce il terzo tipo di elemento che deve
agganciare e la terza coverage a cui deve agganciare	
Sel all	/*seleziona tutti gli archi di aa_ctr
Snap	/*effettua lo snap



Procedura di autocertificazione di un dato

Ambiente operativo: Arc/Info 7

Vengono fornite le seguenti procedure:

“controllo.aml”

“scrivi.aml”.

Queste procedure per funzionare correttamente vanno copiate nella directory di lavoro dove si trova la coverage da certificare.

Per avviare la procedura di autocertificazione lanciare da ambiente “arc” il comando

&r controllo <nome coverage>

I risultati dell'autocertificazione vengono scritti in un file <nome coverage.txt> creato automaticamente dalla procedura nella directory di lavoro stessa.



Controllo.aml

La procedura esegue nell'ordine i seguenti controlli:

- esistenza coverage;
- controllo della precisione;
- se la coverage contiene elementi puntuali
 - numero di punti;
 - tabella pat puntuale (nome e definizione dei campi);
- se la coverage contiene elementi lineari
 - numero di archi;
 - tabella aat lineare (nome e definizione dei campi);
- se la coverage contiene elementi poligonali
 - numero di poligoni;
 - tabella pat poligonale (nome e definizione dei campi);
 - se è corretta la topologia poligonale;
- se la coverage contiene annotation
 - numero di annotation;
- se la coverage contiene regions
 - numero e lista dei nomi delle regions;
 - tabella delle patsubclass delle regions (nome e definizione dei campi);
 - se è corretta la topologia delle regions;
- se la coverage contiene routes
 - numero e lista dei nomi delle routes;
 - tabella delle rat-subclass delle routes (nome e definizione dei campi);
- parametri clean (dangle e fuzzy);
- nodeerrors e labelerrors.

```
/*input argomenti, esistenza coverage e describe
```

```
&args .cov
```

```
&messages &off &all
```

```
&r scrivi NOME COVERAGE: %.cov%
```

```
&if [exists %.cov% -cover] &then &do
```

```
&describe %.cov%
```

```
/*controllo della precisione
```

```
&r scrivi Precisione %dsc$precision%
```

```
/*descrizione dei dati
```

```
/*numero di punti
```

```
&if [exists %.cov% -point] &then &do
```

```
&s punti %dsc$points%
```

```
&r scrivi *****
```

```
&r scrivi PUNTI: %punti%
```

```
&end
```




/*pat puntuale

```
&if [exist %.cov% -point] &then &do
  &if [exist items.txt -file] &then &type [delete items.txt]
  &r scrivi Lista dei campi contenuti della PAT puntuale
  &s val [listitem %.cov% -point items.txt]
  &s nume [open items.txt openstatus -read]
  &r scrivi Legenda dei parametri dei campi:
  &r scrivi <NOME DEL CAMPO> <Larghezza campo, Larghezza di output campo, Tipo di campo,
Numero di decimali>
  &s item [read %nume% readstatus]
  &do &while %readstatus% = 0
    &r scrivi %item% [iteminfo %.cov% -point %item%]
    &s item [read %nume% readstatus]
  &end
  &s val [close -all]
  &s val [delete items.txt]
&end
```

/*numero di archi

```
&if [exists %.cov% -arc] &then &do
  &s archi %dsc$arcs%
  &r scrivi *****
  &r scrivi ARCHI: %archi%
&end
```

/*aat lineare

```
&if [exist %.cov% -arc] &then &do
  &if [exist items.txt -file] &then &type [delete items.txt]
  &r scrivi Lista dei campi contenuti della AAT lineare
  &s val [listitem %.cov% -arc items.txt]
  &s nume [open items.txt openstatus -read]
  &r scrivi Legenda dei parametri dei campi:
  &r scrivi <NOME DEL CAMPO> <Larghezza campo, Larghezza di output campo, Tipo di campo,
Numero di decimali>
  &s item [read %nume% readstatus]
  &do &while %readstatus% = 0
    &r scrivi %item% [iteminfo %.cov% -arc %item%]
    &s item [read %nume% readstatus]
  &end
  &s val [close -all]
  &s val [delete items.txt]
&end
```




/*numero di poligoni

```
&if [exists %cov% -polygon] &then &do
  &s poligoni %dsc$polygons%
  &r scrivi *****
  &r scrivi POLIGONI: %poligoni%
&end
```

/*pat poligonale

```
&if [exist %cov% -polygon] &then &do
  &if [exist items.txt -file] &then &type [delete items.txt]
  &r scrivi Lista dei campi contenuti della PAT poligonale
  &s val [listitem %cov% -polygon items.txt]
  &s nume [open items.txt openstatus -read]
  &r scrivi Legenda dei parametri dei campi:
  &r scrivi <NOME DEL CAMPO> <Larghezza campo, Larghezza di output campo, Tipo di campo,
Numero di decimali>
  &s item [read %nume% readstatus]
  &do &while %readstatus% = 0
    &r scrivi %item% [iteminfo %cov% -polygon %item%]
    &s item [read %nume% readstatus]
  &end
  &s val [close -all]
  &s val [delete items.txt]
&end
```

/*controlli sulla pat per poligoni

```
&if [exist %cov% -polygon] &then &do
&r scrivi *****
&s topo %dsc$topology_polygons%
```

/* se è zero non ha poligoni

```
&if %topo% = 0 &then &do
  &r scrivi ATTENZIONE: La coverage non contiene poligoni
&end
```

/* se è uno ha la topologia prelinare poligonale

```
&if %topo% = 1 &then &do
  &r scrivi ATTENZIONE: coverage modificata e topologia poligonale non ricostruita
&end
```

/*se è due ha la topologia poligonale

```
&if %topo% = 2 &then &do
  &r scrivi La coverage ha la topologia poligonale corretta
&end
&end
```



```

/*numero di annotation
&if [exists %.cov% -annotations] &then &do
    &r scrivi *****
    &s annotazioni %dsc$annotations%
    &r scrivi ANNOTATION: %annotazioni%
&end
/*lista e numero dei nomi delle regions
&if [exist %.cov% -region] &then &do
    &r scrivi *****
    &r scrivi NUMERO E LISTA DELLE SOTTOCLASSI DI REGIONS:
    &s max %dsc$regionsubclasses% + 1
    &s conta 1
    &do &until %conta% = %max%
        &s nome [extract %conta% %dsc$regionnames%]
        &s valori [extract %conta% %dsc$regions%]
        &r scrivi %valori% elementi nella region %nome%
        &s conta %conta% + 1
    &end
&end

/*patsubclass delle regions

&if [exist %.cov% -region] &then &do
    &r scrivi *****
    &describe %.cov%
    &s listareg %DSC$regionnames%
    &r scrivi Legenda dei parametri dei campi:
    &r scrivi <NOME DEL CAMPO> <Larghezza campo, Larghezza di output campo, Tipo di campo,
Numero di decimali>
    &label inizio
    &if %listareg% = " &then &goto fine
        &s reg [before %listareg% ,]
        &s listareg [after %listareg% ,]
        &r scrivi Controllo della region %reg%
    /****** ciclo items
        &if [exist items.txt -file] &then &type [delete items.txt]
        &r scrivi Lista dei campi e dei parametri della PATsubclass
        &s val [listitem %.cov% -region.%reg% items.txt]
        &s nume [open items.txt openstatus -read]
        &s item [read %nume% readstatus]
        &do &while %readstatus% = 0
            &r scrivi %item% [iteminfo %.cov% -region.%reg% %item%]
            &s item [read %nume% readstatus]
        &end
        &s val [close -all]
        &s val [delete items.txt]
    /******
    &goto inizio
    &label fine
&end
    
```



/*controllo regions

```
&if [exist %.cov% -region] &then &do
  &s max %dsc$regionsubclasses% + 1
  &s conta 1
&do &until %conta% = %max%
  &s nome [extract %conta% %dsc$regionnames%]
  &s valori [extract %conta% %dsc$topology_regions%]
  &r scrivi Per la sottoclasse %nome%
  &if %valori% = 0 &then &do
    &r scrivi ATTENZIONE: La coverage non contiene regions
  &end
  &if %valori% = 1 &then &do
    &r scrivi ATTENZIONE: coverage modificata e topologia delle region non ricostruita
  &end
  &if %valori% = 2 &then &do
    &r scrivi La region %nome% ha la topologia corretta
  &end
  &s conta %conta% + 1
&end
&end
```

/*regioerrors

```
&if [exist %.cov% -region] &then &do
  &messages &off &all
&if [exist %.cov% -region] &then &do
  &s max %dsc$regionsubclasses% + 1
  &s conta 1
  &do &until %conta% = %max%
    &s nome [extract %conta% %dsc$regionnames%]
  ae
  disp 9999 0
  edit %.cov%
  ef region.%nome%
  sel unclosed
  &s valuta [show number select]
  q
    &if %valuta% = 0 &then &do
      &r scrivi La region %nome% non contiene elementi aperti
    &end
    &if %valuta% <> 0 &then &do
      &r scrivi La region %nome% contiene %valori% elementi aperti
    &end
    &s conta %conta% + 1
  &end
&end
&messages &on
&end
```

/*lista e numero dei nomi delle routes



```

&if [exist %.cov% -route] &then &do
&r scrivi *****
&r scrivi NUMERO E LISTA DELLE ROUTE SYSTEMS:
&s max %dsc$routesystems% + 1
&s conta 1
&do &until %conta% = %max%
&s nome [extract %conta% %dsc$routenames%]
&s valori [extract %conta% %dsc$routes%]
&r scrivi %valori% elementi nella routes %nome%
&s conta %conta% + 1
&end
&end

/*rat-subclass delle routes
&if [exist %.cov% -route] &then &do
&describe %.cov%
&s listaroute %DSC$routenames%
&r scrivi Legenda dei parametri dei campi:
&r scrivi <NOME DEL CAMPO> <Larghezza campo, Larghezza di output campo, Tipo di campo,
Numero di decimali>
&label inizio1
&if %listaroute% = " &then &goto fine1
&s route [before %listaroute% ,]
&s listaroute [after %listaroute% ,]
&r scrivi Controllo della route %route%
/*ciclo items
&if [exist items.txt -file] &then &type [delete items.txt]
&r scrivi Lista dei campi e dei parametri della RATsubclass
&s val [listitem %.cov% -route.%route% items.txt]
&s nume [open items.txt openstatus -read]
&s item [read %nume% readstatus]
&do &while %readstatus% = 0
&r scrivi %item% [iteminfo %.cov% -route.%route% %item%]
&s item [read %nume% readstatus]
&end
&s val [close -all]
&s val [delete items.txt]
/******
&goto inizio1
&label fine1

&end

/*controllo parametri clean (dangle e fuzzy)
&r scrivi *****

&r scrivi Valore della Tolleranza "DANGLE DISTANCE" %dsc$dangle%

&r scrivi Valore della Tolleranza "FUZZY TOLERANCE" %dsc$fuzzy%
    
```



```

/*controllo nodeerrors e labelerrors
&r scrivi *****
/*nodeerrors
ae
display 9999 0
edit %.cov%
ef arc
sel dangle
&s dangle [show number select]
q
&r scrivi NUMERO ARCHI APPESI (DANGLE NODE) %dangle%

```

```

/*labelerrors
&s topo %dsc$topology_polygons%
&if %topo% = 2 &then &do
&r scrivi ETICHETTE MANCANTI
&watch %.cov%.txt &append
&messages &on
labelerrors %.cov%
&messages &off
&watch &off
&end

```

```

/*****
&end
&else
&type La coverage non esiste.
&messages &on
&return

```

Scrivi.aml

Questa procedura scrive nel file di output i risultati dell'autocertificazione.

```

&args testo:rest
&messages &off &all
&s .ident [open %.cov%.txt opstat -append]

&s wrlog [write %.ident% %testo%]

&type [close %.ident%]
&messages &on
&return

```




Esempio di applicazione della procedura di Autocertificazione.

Riportiamo di seguito un esempio di applicazione della procedura “controllo.aml” applicata ad una coverage di nome “ORIGINE” (che è per brevità una porzione del MOSPAR)

Comando:

&r controllo dati\origine (poichè la coverage è dentro la cartella “dati”)

La procedura salva come risultato, all’interno della cartella contenente la coverage (“dati” nel nostro esempio), il file di testo “origine.txt” con il seguente contenuto:

```
NOME COVERAGE: dati\origine
Precisione DOUBLE
*****
POLIGONI: 33
Lista dei campi contenuti della PAT poligonale
Legenda dei parametri dei campi:
<NOME DEL CAMPO> <Larghezza campo, Larghezza di output campo, Tipo di campo, Numero di decimali>
valori univoci: .....
AREA 8,18,F,5
PERIMETER 8,18,F,5
ORIGINE# 4,5,B,0
ORIGINE-ID 4,5,B,0
AREA_TUT 1,1,I,0
valori univoci: 5
CODICE_PA 2,3,B,0
valori univoci: 1
CODICE_RI 2,3,B,0
valori univoci: 0
CODICE_PL 2,3,B,0
valori univoci: 0
PAR_NAT 1,1,I,0
valori univoci: 1,2
CODICE_MN 2,3,B,0
valori univoci: 0
RISER 2,2,C,0
valori univoci: 99,t2,t4
AGRIC 2,2,C,0
valori univoci: 99,a1,a2,a3
TUT 2,2,C,0
valori univoci: 99,t5
FRUIZ 2,2,C,0
valori univoci: 99,f2
DEGRADI 2,2,C,0
valori univoci: 99,p3
MONUM 2,2,C,0
valori univoci: 99
PA_PRIV 2,2,C,0
valori univoci: 99
PAR_AL 2,2,C,0
valori univoci: 99
AZ_CO 2,2,C,0
valori univoci: 99,c1
MON_NAT 2,2,C,0
valori univoci: 99
NO_INFO 2,2,C,0
valori univoci: 99
SPECCHI 2,2,C,0
valori univoci: 99,s1
INFRA 2,2,C,0
valori univoci: 99,i1
COD_AREA 4,5,B,0
COD_AREA:
Superato il limite di 30 elementi univoci individuati.
Non si ritiene opportuno riportarne i valori.
*****
La coverage ha la topologia poligonale corretta
*****
NUMERO E LISTA DELLE SOTTOCLASSI DI REGIONS:
0 elementi nella region MONUM
0 elementi nella region PAR_LOC
2 elementi nella region PAR_NAT
1 elementi nella region PAR_RENA
25 elementi nella region PMOSAICO
0 elementi nella region RI_RENA
*****
Legenda dei parametri dei campi:
<NOME DEL CAMPO> <Larghezza campo, Larghezza di output campo, Tipo di campo, Numero di decimali>
Controllo della region MONUM
Lista dei campi e dei parametri della PATsubclass
AREA 8,18,F,5
PERIMETER 8,18,F,5
MONUM# 4,5,B,0
MONUM-ID 4,5,B,0
CODICE_MN 2,3,B,0
valori univoci: ''
Controllo della region PAR_LOC
Lista dei campi e dei parametri della PATsubclass
AREA 8,18,F,5
PERIMETER 8,18,F,5
```



PAR_LOC# 4,5,B,0
 PAR_LOC-ID 4,5,B,0
 CODICE_PL 2,3,B,0
 valori univoci: ''

Controllo della region PAR_NAT
 Lista dei campi e dei parametri della PATsubclass
 AREA 8,18,F,5
 PERIMETER 8,18,F,5
 PAR_NAT# 4,5,B,0
 PAR_NAT-ID 4,5,B,0
 CODICE_PA 2,3,B,0
 valori univoci: 1
 PAR_NAT 1,1,1,0
 valori univoci: 1,2

Controllo della region PAR_RENA
 Lista dei campi e dei parametri della PATsubclass
 AREA 8,18,F,5
 PERIMETER 8,18,F,5
 PAR_RENA# 4,5,B,0
 PAR_RENA-ID 4,5,B,0
 CODICE_PA 2,3,B,0
 valori univoci: 1

Controllo della region PMOSAICO
 Lista dei campi e dei parametri della PATsubclass
 AREA 8,18,F,5
 PERIMETER 8,18,F,5
 PMOSAICO# 4,5,B,0
 PMOSAICO-ID 4,5,B,0
 MOS_PAR 2,2,C,0
 valori univoci: a1,a2,a3,c1,f2,i1,p3,s1,t2,t4,t5

Controllo della region RI_RENA
 Lista dei campi e dei parametri della PATsubclass
 AREA 8,18,F,5
 PERIMETER 8,18,F,5
 RI_RENA# 4,5,B,0
 RI_RENA-ID 4,5,B,0
 CODICE_RI 2,3,B,0
 valori univoci: ''

Per la sottoclasse MONUM
 La region MONUM ha la topologia corretta
 Per la sottoclasse PAR_LOC
 La region PAR_LOC ha la topologia corretta
 Per la sottoclasse PAR_NAT
 La region PAR_NAT ha la topologia corretta
 Per la sottoclasse PAR_RENA
 La region PAR_RENA ha la topologia corretta
 Per la sottoclasse PMOSAICO
 La region PMOSAICO ha la topologia corretta
 Per la sottoclasse RI_RENA
 La region RI_RENA ha la topologia corretta
 La region MONUM non contiene elementi aperti
 La region PAR_LOC non contiene elementi aperti
 La region PAR_NAT non contiene elementi aperti
 La region PAR_RENA non contiene elementi aperti
 La region PMOSAICO non contiene elementi aperti
 La region RI_RENA non contiene elementi aperti

NUMERO E LISTA DELLE ROUTE SYSTEMS:

0 elementi nella routes PAR_LOC
 1 elementi nella routes PAR_NAT
 1 elementi nella routes PAR_RENA
 0 elementi nella routes RI_RENA

Legenda dei parametri dei campi:

<NOME DEL CAMPO> <Larghezza campo, Larghezza di output campo, Tipo di campo, Numero di decimali>

Controllo della route PAR_LOC
 Lista dei campi e dei parametri della RATsubclass
 PAR_LOC# 4,5,B,0
 PAR_LOC-ID 4,5,B,0
 PERIMETRO 1,1,1,0
 valori univoci: ''

Controllo della route PAR_NAT
 Lista dei campi e dei parametri della RATsubclass
 PAR_NAT# 4,5,B,0
 PAR_NAT-ID 4,5,B,0
 PERIMETRO 1,1,1,0
 valori univoci: 1

Controllo della route PAR_RENA
 Lista dei campi e dei parametri della RATsubclass
 PAR_RENA# 4,5,B,0
 PAR_RENA-ID 4,5,B,0
 PERIMETRO 1,1,1,0
 valori univoci: 1

Controllo della route RI_RENA
 Lista dei campi e dei parametri della RATsubclass
 RI_RENA# 4,5,B,0
 RI_RENA-ID 4,5,B,0
 PERIMETRO 1,1,1,0
 valori univoci: ''

 Valore della Tolleranza "DANGLE DISTANCE" 0.000000001
 Valore della Tolleranza "FUZZY TOLERANCE" 0.0001

NUMERO ARCHI APPESI (DANGLE NODE) 0
 ETICHETTE MANCANTI
 Polygon 1 has 0 label points.
 Total number of Polygons with No Labels: 1
 Total number of Polygons with Multiple Labels: 0





Software di misura e certificazione della Congruenza esterna:

Ambiente operativo: Arc/Info 7

Sono state realizzate due procedure in ambiente Arc/Info per il software di misura della congruenza.

Cong1.aml

Questa procedura, basata sul comando NEAR di ArcInfo documenta, con un apposito strato informativo e un file di testo, il numero di vertici coincidenti dello strato informativo da controllare con quelli della base topografica di riferimento scelta o giacenti lungo segmenti di quest'ultima.

```
&args strato base risultato /*strato è la coverage da controllare, base è la base sulla quale
controllare (schacciata)
precision double double
copy %base% base
build base line
copy %strato% strato
build strato line
arcpoint strato strato_p line strato-id 0.1
kill strato all
near strato_p base line 50 ris location
kill (!base strato_p!) all
```

```
&describe ris
&s punti_tot %DSC$points%
reselect ris %risultato% point # point
reselect Distance <= 5 and Distance > 0.00001 and Base# <> 0/* Questi sono i vertici non congruenti
[unquote "]
n
n
&describe %risultato%
&s punti_err %DSC$points%
reselect ris punti_ok point # point
reselect Distance <= 0.00001 and Base# <> 0/*Questi sono i vertici coincidenti
[unquote "]
n
n
```

```
kill ris all
```

```
&describe punti_ok
&s punti_ok %DSC$points%
kill punti_ok all
&s perc_nok [calc [calc %punti_err% / %punti_tot%] * 100]
&s perc_ok [calc [calc %punti_ok% / %punti_tot%] * 100]
```



```
&s riga1 [quote Vertici totali:      %punti_tot%]  
&s riga2 [quote Vertici non congruenti: %punti_err%]  
&s riga3 [quote Vertici coincidenti:  %punti_ok%]  
&s riga4 [quote Percentuale possibili vertici non congruenti: %perc_nok%]  
&s riga5 [quote Percentuale vertici coincidenti:      %perc_ok% ]
```

```
&s file %risultato%.txt  
&s fileunit = [open %file% openstatus -write]  
&s riga [write %fileunit% %riga1%]  
&s riga [write %fileunit% %riga2%]  
&s riga [write %fileunit% %riga3%]  
&s riga [write %fileunit% %riga4%]  
&s riga [write %fileunit% %riga5%]  
&s chiudi [close %fileunit%]
```

```
&sys type %risultato%.txt
```



Cong2.aml

Questa procedura basata sul concetto di micropoligono, serve a contare e segnalare i possibili errori venutisi a creare a causa di incongruenze fra strato da controllare e base: si consiglia di utilizzare questa procedura una volta avuti risultati soddisfacenti dalla prima. Come vedremo infatti, lo strato informativo creato è un'utile guida per individuare e correggere gli errori tra quelli riportati.

```

/* Preparare prima la base (schiacciata) con topologia poligonale
&args strato base risultato
precision double double
copy %base% base
pullitems base.pat base.pat /* Si fa un po' di pulizia fra gli Items
Area
Perimeter
base#
base-id
end
copy %strato% strato
pullitems strato.pat strato.pat /* Pulizia
Area
Perimeter
strato#
strato-id
end
additem base.pat base.pat base 2 2 i /*Questi Items serviranno per distinguere lo strato dalla base dopo la
Union
additem strato.pat strato.pat strato 2 2 i
tables
sel base.pat
res $recno <> 1
calc base = 1
sel strato.pat
res $recno <> 1
calc strato = 1
quit
union strato base ris 0.0001
kill (!base strato!) all
pullitems ris.pat ris.pat /* Pulizia
area
perimeter
ris#
ris-id
base#
strato#
base
strato
end
    
```



```
frequency ris.pat ris.frq /* Quanti elementi della base hanno internal number ripetuto?
base# /* Questo serve ad evitare che micro-poligoni già esistenti nella base
end /* vengano calcolati come errori.
end
dropitem ris.frq ris.frq case#
tables
sel ris.frq
sort base#
q
joinitem ris.pat ris.frq ris.pat base# base# ordered
tables
$kill ris.frq
q
reselect ris %risultato% poly # poly
reselect area < 50 and frequency > 1 and base# <> 1/* Seleziono tutte le aree minori di 50 Mq che hanno
incrementato gli elementi della base
[unquote "]
n
n
kill ris all
&return
```



Operazioni da eseguire con ArcInfo per rendere congruenti ad una base i dati già acquisiti o da acquisire.

Ambiente operativo: Arc/Info 7

Reshape

Per questo ambiente operativo è stata realizzata una procedura (reshape.aml) per agevolare le operazioni di editing manuale mirate alla risoluzione dei problemi di congruenza fra coverages.

Requisiti per il corretto funzionamento della procedura:

1. disporre delle seguenti:
 - coverage di base (es: schiacciata)
 - coverage congruenda (es: ottenuta dall'intersezione della coverage da correggere con la schiacciata) ovvero quella che verrà editata
 - coverage serbatoio di attributi (copia della coverage da correggere)
2. specificare il campo codice ed aggiungere con il comando *additem* il campo "stato" (di tipo carattere di almeno 3 caratteri) alle coverages congruenda e serbatoio

Per lanciare la procedura utilizzare il comando:

```
&r reshape <nome coverage di base> <coverage congruenda> <campo codice> <coverage serbatoio>
```

<nome coverage di base> /*il nome deve essere completo del percorso interno che individua la directory nella quale è contenuta la coverage di base

<coverage congruenda> /*il nome deve essere completo del percorso interno che individua la directory nella quale è contenuta la coverage congruenda

<campo codice> /*nome dell'item contenente un codice significativo

<coverage serbatoio> /*nome della coverage (copia dell'originale da rendere congruente) dalla quale prelevare gli attributi andati persi durante le operazioni di editing


All'interno del listato della procedura reshape.aml è possibile settare i parametri di tolleranza.

Riportiamo di seguito uno schema gerarchico che rappresenta la struttura degli elementi facenti parte della procedura e relativi listati.



ARGOMENTI:


<NOME COVERAGE DI BASE>
<COVERAGE CONGRUENDA>
<CAMPO CODICE>
<COVERAGE SERBATOIO>



RESHAPE.AML:

- ambiente editing lineare
- tolleranze

lancia "barra.menù"




BARRA.MENU

gestisce "Reshape menu" per l'editing lineare

BARRA.AML


Per rilanciare il
"reshape" da ArcEdit



SET.AML

ambiente editing poligonale

Lancia il menu "Scegli"



BARRA1.MENU

gestisce il menu "Scegli" per l'editing poligonale

AP_MOSPAR

Gestisce i colori

ATT2.AML

copia gli attributi
dalla coverage
serbatoio



Reshape.aml

```

/*****
/*QUESTA PROCEDURA PRENDE IN INPUT
/*COVERAGE BASE, COVERAGE CONGRUENDA, NOME-CAMPO-CODICE, COVERAGE
SERBATOIO
/*IMPOSTA L'AMBIENTE DI EDITING
/*CONSENTE DI PERSONALIZZARE I VALORI DELLE TOLLERANZE
/*RICHIAMA BARRA.MENU
/*****
&args .orig .destin .cod .serb
&if ( %.orig% = '' ) or ( %.destin% = '' ) or ( not [Exist %.orig% -cover] ) or ( not [Exist %.destin% -cover]
) &then &do
  &terminal 9999
  &menu attenzione.menu &position &above &display &stripe 'Inserimento Coverages di Origine e
Destinazione'
&end
&if [quote %.orig%] = '0' &then &stop
&s .orig [locase %.orig%]
&s .destin [locase %.destin%]

/*****IMPOSTA L'AMBIENTE DI EDITING DI ARCEDIT
precision double double
ae
disp 9999 1
edit %.orig% arc
drawe arc lab
drawe node dangle
nodecolor dangle 2
bc %.destin% 4
be arc
draw
/***** TOLLERANZE *****/
arcsnap on 5
nodesnap closest 5
editdistance 8
/*****

&terminal 9999
&menu barra.menu &sidebar &position &right &display &stripe 'Reshape Menu'
&return
    
```




Barra.menu

/*Inizializza il menu

2

Accorpa_poly &r set.aml

Preleva_arco &type 'SELEZIONA 'L'ARCO";sel;put %.destin%;y;edit %.destin% arc;bc %.orig% 3;be arc;draw;sel;delete;edit %.orig% arc;draw

Un_taglio

Appeso sel;split;sel;put %.destin%;y;edit %.destin% arc;bc %.orig% 3;be arc;draw

Non_appeso sel;split;sel;put %.destin%;y;edit %.destin% arc;bc %.orig% 3;be arc;draw;sel;split;sel;delete;draw

Due_tagli sel;split;sel;split;sel;put %.destin%;y;edit %.destin% arc;bc %.orig% 3;be arc;draw;sel;split;sel;split;sel;delete;draw

Copia_arco put %.destin%;y

Origine edit %.orig% arc;bc %.destin% 4;be arc;draw

Destinazione edit %.destin% arc;bc %.orig% 3;be arc;draw

Seleziona

Sel_Uno sel

Sel_Più sel many

Sel_Polig sel poly

Sel_Rettang sel box

Deseleziona unsel all

Estendi sel;extend

Spezza split

Elimina delete;draw

Ridisegna draw

Salva

Salva save

Salva_Tutto save all;y

Annulla oops

ArcEdit &return



Barra.aml

*/*attiva la barra dei menu' da arcedit*

&terminal 9999

&menu barra.menu &sidebar &position &right &display &stripe 'Reshape Menu'

&return



Set.aml

/*Editing dati da rendere congruenti alla CTR

/*Questa procedura consente di impostare l'ambiente di editing ed attivare la barra del menu' "Scegli"

edit %.destin%

de poly

ef poly

ap ap_mospar

textitem %.destin% label %.cod%

draw

&terminal 9999

&menu barra1.menu &sidebar &position &left &display &stripe 'Scegli'

&return



Ap_mospar.aml

/*Evidenzia gli archi della base in background disegnandoli in blu con uno spessore maggiore.
/*Colora i poligoni della coverage congruenda che hanno l'item stato codificato con il valore "ok"

```
&type [show lineset]
lineset plotter
linesymbol 1
linesize 0.1
linecolor 'blue'
arcs %.orig%
```

```
shadeset color
reselect %.destin% poly stato = 'ok'
polygonshade %.destin% 11
aselect %.destin% poly stato = 'ok'
polygonshade %.destin% 11
```

```
lineset color
linesymbol 1
linesize 0.1
```

```
&return
```



Barra1.menu

/*Imposta il menu per l'editing poligonale

2

Seleziona

Sel_one sel
 Sel_many sel many
 sel_poly sel poly
 Sel_box sel box

Deseleziona unsel all

Merge merge

Merge_selezionati merge select

Visualizza

Campo_cod drawe lab;textitem %.destin% lab %.cod%;draw

Off_cod drawe lab off;draw

/* Campo_strato textitem %.destin% arc strato_ctr;drawe arc;draw

/* Off_strato textitem %.destin% clear;drawe arc;draw

Split split

Delete delete

Attributi &r att2

Build Build

Codifica calculate stato = 'ok';save;draw

Draw draw

Save save

Oops oops

Esci ap none;draw;&return



Att2.aml

```

/*Opera nell'ambiente di lavoro di ARCEDIT.
/*Procedura che copia gli attributi associati ad un elemento di una coverage ad un altro elemento di un'altra coverage.
/*E' necessario selezionare l'elemento di cui si vogliono aggiornare gli attributi
/*Lanciare la procedura con il comando "&run att"
/*Selezionare l'elemento della coverage serbatoio dal quale si vogliono prelevare gli attributi
/*Deve essere selezionato un solo elemento
    
```

```

&type [close -all]
/***** CONTROLLA CHE SIA SELEZIONATO UN SOLO POLIGONO
ef poly
&s selez [show number select]
&if %selez% = 0 &then &do
    &type ATTENZIONE: selezionare almeno un poligono prima di lanciare la procedura
    &return
&end

&if %selez% > 1 &then &do
    &type ATTENZIONE: deve essere selezionato un solo poligono prima di lanciare la procedura
    &return
&end

/***** MEMORIZZA NELLA VARIABILE num IL RIFERIMENTO AL POLIGONO
SELEZIONATO
&s num [show select 1]

/***** EDITA LA COVERAGE SERBATOIO E PRELEVA GLI ATTRIBUTI
edit %.serb%
ef poly
&type SELEZIONARE L'ELEMENTO DAL QUALE SI VUOLE PRELEVARE GLI ATTRIBUTI
sel

&s num_sel [show number select]
&if %num_sel% = 0 &then &do
    &type ATTENZIONE: La procedura e' stata fermata perche' non e' stato selezionato alcun elemento.
    &type Controllare che nel punto dove si e' effettuato il click ci sia un elemento.
    &type Non e' stata effettuata alcuna modifica dei dati.
    edit %.destin%
    &return
&end

&s numitems [show number items]

&s numero [show select 1]
    
```



&s conta 5 /*esclude area, perimeter, # e ID

&label inizioconta

&s nome%conta% [show items %conta% name]

&s tipo%conta% [extract 4 [show items %conta% definition]]

&s conten%conta% [show polygon %numero% item [show items %conta% name]]

&s conta [calc %conta% + 1]

&if %conta% <= %numitems% &then &goto inizioconta

/****** EDITA LA COVERAGE DESTINAZIONE ED ASSEGNA GLI ATTRIBUTI AL POLIGONO
GIA' SELEZIONATO

edit %.destin%

ef poly

sel \$recno = %num%

&type ATTENDERE PREGO....

&s conta 5

&label inizioconta2

&if [value tipo%conta%] = 'C' &then

calc [value nome%conta%] = [quote [value conten%conta%]]

&else

calc [value nome%conta%] = [value conten%conta%]

&s conta [calc %conta% + 1]

&if %conta% <= %numitems% &then &goto inizioconta2

draw

&return



Altre Procedure

Crea_schiacciata.aml

Questa procedura a partire da un file (di nome: "lista.txt") contenente la lista degli strati informativi di partenza, crea la schiacciata. Le coverages di partenza devono risiedere in un'unica directory. Il file "lista.txt", contenente il nome delle coverages di partenza deve finire con un "ritorno carrello". Viene creata la schiacciata con intersezioni all'incrocio degli archi. Si é scelto di implementare la clean con una tolleranza di 1 millimetro. Le schiacciate create sono di tipo lineare

```

&s barra \
&s input
&s out_dir
&s schiacciata schiacciata
&s lista lista.txt
&s lavoro lavoro

/* Si copiano le coverages nella directory di lavoro
/* Si crea la topologia lineare e si uniformano gli attributi
/* per la successiva APPEND. Si aggiunge agli attributi il nome
/* della coverage di partenza.
precision double double
cw %lavoro%
&s output %schiacciata%
&s file_read [open %lista% readstat -r]
&s done .TRUE.
&do &while %done%
&s nome_cov [read %file_read% readstat]
&if [null %nome_cov%] &then &goto fine
&s out_cov %lavoro%%barra%%nome_cov%
copy %nome_cov% %out_cov%
build %out_cov% line
pullitems %out_cov%.aat %out_cov%.aat
fnode#
tnode#
lpoly#
rpoly#
length
%nome_cov%#
%nome_cov%-id
end

additem %out_cov%.aat %out_cov%.aat Strato 100 100 c
tables %lavoro%%barra%info
sel %nome_cov%.aat
move [quote %nome_cov%] to strato
q
    
```



```
&end
&label fine
&type [close -all]
```

```
append %output% arc all
&s file_read [open %lista% readstat -r]
&s done .TRUE.
&do &while %done%
&s nome_cov [read %file_read% readstat]
&if [null %nome_cov%] &then &goto fine1
%lavoro%%barra%%nome_cov%
&end
&label fine1
&type [close -all]
end
```

```
&s file_read [open %lista% readstat -r]
&s done .TRUE.
&do &while %done%
&s nome_cov [read %file_read% readstat]
&if [null %nome_cov%] &then &goto fine2
```

```
kill %lavoro%%barra%%nome_cov% all
```

```
&end
&label fine2
&type [close -all]
end
```

```
tables info
sel %schiacciata%.aat
calc %schiacciata%-id = %schiacciata%#
q
idedit %output% line
clean %output% %output%_i 0 0.001 line
kill %output% all
rename %output%_i %schiacciata%
dw lavoro
y
&return
```



Fpg_in_Sc_line.aml

Questa procedura crea un nuovo dato lineare "sc_line_attr" a partire da un dato lineare %linea% e da un dato poligonale %poligono%.

Viene conservata la geometria del dato %linea% con, riportate in tabella .aat, le informazioni del poligono destro e sinistro provenienti dal dato %poligono%

```

&args linea poligono
precision double double
identity %linea% %poligono% sc_line_attr line 0.000001
dropitem sc_line_attr.aat sc_line_attr.aat /* qui vanno lasciati i campi significativi
area
perimeter
%poligono%-id
%linea%#
%linea%-id
%poligono%#
codice
end
additem sc_line_attr.aat sc_line_attr.aat %poligono%# 4 5 b
tables
sel sc_line_attr.aat
calc %poligono%# = lpoly#
q
joinitem sc_line_attr.aat %poligono%.pat sc_line_attr.aat %poligono%# %poligono%# ordered
dropitem sc_line_attr.aat sc_line_attr.aat /* qui vanno lasciati i campi significativi
area
perimeter
%poligono%#
%poligono%-id
end
tables
sel sc_line_attr.aat
alter
codice
codice_1
[unquote "]
[unquote "]
[unquote "]
[unquote "]
end
additem sc_line_attr.aat sc_line_attr.aat %poligono%# 4 5 b
tables
sel sc_line_attr.aat
calc %poligono%# = rpoly#
q
    
```



```
joinitem sc_line_attr.aat %poligono%.pat sc_line_attr.aat %poligono%# %poligono%# ordered
dropitem sc_line_attr.aat sc_line_attr.aat
area
perimeter
%poligono%#
%poligono%-id
end
tables
sel sc_line_attr.aat
alter
codice
codice_r
[unquote "]
[unquote "]
[unquote "]
[unquote "]
end
```



Esempio 1:

Data una coverage (Dato_Part) da rendere congruente con una base costituita da diversi strati (P1, P2,... L1, L2....) costruiamo con il comando APPEND (di Arc/Info) le coverages:

FPG (opzione poly)

Sc_Line_Full (opzione line)

e con la procedura "crea_schiacciata.aml" la coverage:

Sc_DP_LinePoly

Inoltre per poter applicare correttamente la procedura "reshape.aml" sopra esposta si deve considerare come

<nome coverage di base> la coverage **Sc_line_full**

<coverage congruenda> la coverage **Sc_DP_LinePoly**

<coverage serbatoio> una copia della coverage **Sc_DP_LinePoly**

Di seguito sono riportate alcune spiegazioni di come utilizzare i tools forniti in ambiente Arc/Info nella procedura "reshape.aml".




Figura 1: situazione di partenza

Funzionalità fornite dalla procedura Reshape.aml.

Funzione preleva arco:

Avviare reshape.aml inserendo gli opportuni argomenti

Ossia:
`&r reshape <nome coverage di base> <coverage congruenda>
 <campo codice> <coverage serbatoio>`

<nome coverage di base> nome completo del percorso interno che individua la directory nella quale è contenuta la coverage di base
 <coverage congruenda> nome completo del percorso interno che individua la directory nella quale è contenuta la coverage congruenda
 <campo codice> nome dell'item contenente un codice significativo
 <coverage serbatoio> nome della coverage (copia dell'originale da rendere congruente) dalla quale prelevare gli attributi andati persi durante le operazioni d'editing

Utilizzando le opportune funzionalità di Zoom e Pan per individuare le zone da

correggere si arrivi per esempio a visualizzare la situazione rappresentata in Figura 1. La coverage da correggere è rappresentata in blu mentre la coverage di base è in editing ed è bianca. In questo caso per ristabilire la congruenza fra le due coverage, occorre copiare il tratto d'arco bianco a destra della label ed eliminare l'analogo tratto blu della coverage congruenda. E' possibile velocizzare questa

operazione, utilizzando il bottone **Preleva_arco**, dal "Reshape menù" (vedi figura 2), che consente di selezionare l'arco da prelevare (vedi figura 3) selezionare l'arco da eliminare (vedi figura 4) per ottenere automaticamente il risultato visualizzato in figura 5.



Figura 2: Reshape menù




Figura 3: selezione arco da copiare




Figura 4: selezione arco da eliminare




Figura 5: risultato



Bottone “un taglio”:

Si supponga di dover prelevare un tratto d’arco per il quale occorra effettuare un taglio.

E’ possibile velocizzare questa operazione, utilizzando il bottone **Un_taglio**:

nella coverage di base consente di:

- selezionare l’arco in cui è contenuto il tratto d’arco da prelevare
- indicare con un click il punto dove deve essere spezzato l’arco
- selezionare il tratto d’arco da prelevare

a questo punto il tratto d’arco da prelevare viene automaticamente copiato nella coverage congruenda che viene messa in editing, allora:

- selezionare l’arco in cui è contenuto il tratto d’arco da sostituire
- indicare con un click il punto dove deve essere spezzato l’arco
- selezionare il tratto d’arco da eliminare

nota:

se i nodi dell’arco copiato nella coverage congruenda hanno una distanza compatibile con la tolleranza fissata verranno automaticamente “snappati” ai rispettivi elementi più prossimi.

Bottone “due tagli”:

Si supponga di dover prelevare un tratto d’arco per il quale occorra effettuare due tagli.

E’ possibile velocizzare questa operazione, utilizzando il bottone **Due_tagli**:

nella coverage di base consente di;

- selezionare l’arco in cui è contenuto il tratto d’arco da prelevare
- indicare con un click il primo punto dove deve essere spezzato l’arco
- selezionare l’arco in cui è contenuto il tratto d’arco da prelevare per il secondo taglio
- indicare con un click il secondo punto dove deve essere spezzato l’arco
- selezionare il tratto d’arco da prelevare

a questo punto il tratto d’arco da prelevare viene automaticamente copiato nella coverage congruenda che viene messa in editing, allora:

- selezionare l’arco in cui è contenuto il tratto d’arco da sostituire
- indicare con un click il primo punto dove deve essere spezzato l’arco
- selezionare l’arco in cui è contenuto il tratto d’arco da eliminare per il secondo taglio
- indicare con un click il secondo punto dove deve essere spezzato l’arco
- selezionare il tratto d’arco da eliminare

nota:

se i nodi dell’arco copiato nella coverage congruenda hanno una distanza compatibile con la tolleranza fissata verranno automaticamente “snappati” ai rispettivi elementi più prossimi.



Accorpamento di poligoni:




Figura 6: menù "Scegli"

Dopo aver lanciato la procedura di Reshape con gli argomenti necessari, dal menù "Reshape menu" (figura 2) fare click sul bottone "Accorpa poligoni" per accedere al menù "Scegli" (figura 6) e visualizzare in maniera opportuna le coverage:

infatti verranno rappresentate:

- in rosso, la coverage congruenda (solitamente intersezione di coverage non congruente e schiacciata)
- in blu, la coverage di base (ossia la schiacciata)
- in bianco, le etichette e relativi attributi prelevati dal campo codice (uno degli argomenti dati in pasto alla procedura Reshape.aml)

(vedi figura 7)

I micro-poligoni con codifica zero che compaiono in figura 7 sono quelli che evidenziano la discordanza fra base e coverage congruenda. Possono essere agevolmente eliminati utilizzando le funzionalità che descriviamo di seguito.

Il bottone "Merge" consente di eseguire la fusione fra due poligoni:

- Fare click sul bottone "Merge" nel menù "Scegli" (figura 6)
- nella finestra grafica con il primo click selezionare il primo poligono da unire
- con il secondo click selezionare il secondo poligono da unire contenente gli attributi da conservare

i poligoni verranno così automaticamente uniti.




Figura 7: esempio di visualizzazione delle coverage nell'ambiente reshape per l'editig dei poligoni.

Con la combinazione dei bottoni di "Seleziona" con opzione sel many ed il bottone "Merge" è possibile eseguire rapidamente la merge di tutti i poligoni selezionati.

Con il bottone "Attributi" è possibile recuperare gli attributi eventualmente persi durante l'operazione di merge. Per far questo:

- selezionare l'elemento che ha perso gli attributi (bottone "Seleziona" con l'opzione "one")
- con un click sul bottone "Attributi" verrà editata la *coverage serbatoio* dalla quale prelevare gli attributi
- nella finestra grafica fare click

sull'elemento che contiene gli attributi che interessano, automaticamente essi verranno copiati all'interno dell'elemento della coverage congruenda che li aveva persi.



Build

Il bottone “Build” consente di ricostruire la topologia poligonale, che è indispensabile ristabilire quando sono state apportate delle modifiche agli archi della coverage.

Codifica

Il bottone “Codifica” permette di codificare il campo obbligatorio “Stato” con il valore “ok” questo risulta utile quando il poligono su cui si è operato è stato reso congruente e lo si vuole marcare (colorandolo di blu) per evitare di doverlo riesaminare.

Gli altri bottoni che appaiono nei “Reshape menù” e “Scegli” sono una riedizione o una semplice combinazione di comandi d’Arcedit e sono stati inseriti per rendere più agevole l’editing.

Ogni qual volta si rende necessario utilizzare gli altri comandi di ArcEdit è possibile accedervi dal bottone “ArcEdit” del “Reshape menù”. Per poter rientrare da ArcEdit al “Reshape menù” basta digitare il seguente comando:

&r barra

Esempio 2:

Ambiente operativo: ArcView

In quest'esempio utilizzeremo alcune delle funzionalità inserite nell'estensione EDITING TOOLS di ArcView.

Immaginiamo di avere uno strato poligonale da rendere congruente con un'insieme di strati base (lineari e/o poligonali). Poiché lo strato da correggere si appoggerebbe anche su elementi lineari facenti parte degli strati della base, è opportuno prendere i temi che possono interessare e realizzare un unico strato di base lineare che chiameremo **schiacciata**.

Abbiamo così due temi: lo strato da rendere congruente (poligonale) e la schiacciata (lineare)

E' necessario adottare strategie differenti secondo la forma che hanno i dati da rendere congruenti. Ad esempio se l'arco al quale appoggiarsi contiene molti vertici è utile la funzionalità di "Reshape" che descriviamo di seguito.




Figura 1: porzione di strato da rendere congruente



Figura 2: porzione di schiacciata




Figura 3 : porzione di strato da rendere congruente e schiacciata sovrapposti

In figura 1 è rappresentata una porzione di strato da rendere congruente.

In figura 2 è rappresentata una porzione di schiacciata.

In figura 3 sono rappresentati i due temi sovrapposti ed è possibile notare che ci sono evidenti scostamenti fra le linee che rappresentano i due strati.






Figura 4: vertici di archi della schiacciata messi in evidenza

Selezioniamo alcuni archi della schiacciata ed evidenziamo i vertici con lo strumento  (vedi figura 4). Ogni vertice è stato marcato col simbolo 









Figura 5: vertici a confronto

Mettiamo in editing lo strato poligonale ed evidenziamo i vertici del poligono interessato con lo strumento  (vedi figura 5)

Questo evidenzia che i vertici da modificare per portare a congruenza dati e schiacciata sono numerosi, conviene quindi adottare il “metodo di reshape”, invece che quello di snap.

Metodo di reshape:

- **rendere attiva, sulla legenda della View, la schiacciata;**
- **selezionare, con lo strumento  l'arco della schiacciata al quale si vuole adattare il tratto di perimetro dei poligoni da correggere;**
- **scegliere, “Avvia - Reshape”, dal menù “EDITING TOOLS” della vista, per attivare la “fase di Reshape”;**
- **Prendere, lo strumento seleziona in una linea o in un poligono . Quando viene selezionato, il puntatore del mouse diventa della forma  , per segnalare che dal clic successivo si potranno nell'ordine designare:**
- **primo clic sul primo nodo della spezzata che verrà marcato con  ;**




Figura 6: preleva linea del "metodo di reshape"




Figura 7: marcatura del tratto di poligono da modificare







Figura 8: modifiche effettuate dal reshape sul tratto di perimetro selezionato

- secondo clic sul secondo nodo della spezzata che verrà marcato con ● ;
- terzo clic su un qualsiasi vertice interno che verrà marcato con ■ ; (vedi figura 6).
- selezionare "Salva Spezzata" dal menu PopUp della view, così che sarà possibile utilizzare lo stesso tratto di spezzata marcata sulla schiacciata, per modificare i due poligoni adiacenti.
- mettere in editing il tema da correggere;
- con lo strumento  "Vertex Edit", selezionare il poligono da modificare;
- Attivare  seleziona in un poligono, ed eseguire un'operazione analoga a quanto fatto sopra, avendo cura di rispettare l'ordine preciso di marcatura dei vertici utilizzato per la spezzata:
- primo clic primo nodo del poligono, che verrà marcato con ● ;
- secondo clic secondo nodo del poligono, che verrà marcato con ● ;
- terzo clic un qualsiasi vertice interno, che verrà marcato con ■ ; (vedi figura 7).
- Fare clic ora sul bottone  "Esegue incolla spezzata", per completare l'operazione. (Vedi figura 8)




Figura 8 bis: dettaglio modifica poligono




Figura 9: poligono adiacente




Figura 8: modifiche effettuate dal reshape sul tratto di perimetro selezionato











Figura 9: risultato delle modifiche sul poligono adiacente

- con lo strumento  “Vertex Edit”, selezionare il poligono adiacente da modificare;
- Attivare  seleziona in un poligono, ed eseguire nuovamente l’operazione di marcatura dei vertici nel poligono adiacente:
- primo clic primo nodo del poligono, che verrà marcato con  ;
- secondo clic secondo nodo del poligono, che verrà marcato con  ;
- terzo clic un qualsiasi vertice interno, che verrà marcato con  ;
- Fare clic ora sul bottone  “Esegue incolla spezzata”, per completare l’operazione. (Vedi figura 9)

Notare come i vertici del poligono adiacente, per il tratto marcato, si siano spostati per andare a coincidere con il tratto di spezzata e quindi con l’altro poligono. (vedere anche il dettaglio in figura 9 bis).

Con il bottone , eliminate le \times sui vertici e deselezionato il poligono, si ottiene il risultato voluto (vedi figura 10)




Figura 9 bis: particolare del risultato delle modifiche sul poligono adiacente




Figura 10: congruenza ottenuta con il “metodo di reshape”

Esempio 3:

Una possibile metodologia di lavoro può essere quella di creare un unico dato poligonale ottenuto dall'intersezione dello strato da rendere congruente con la schiacciata e procedere per successivi accorpamenti di poligoni; a questi accorpamenti possono seguire o possono essere inframezzate funzioni di reshape o snap per quegli archi appesi che non vengono visualizzati sulla schiacciata poligonale. Consideriamo come dati di partenza già disponibili, la schiacciata lineare (figura 11) la schiacciata poligonale (figura 12) che possiamo vedere entrambe sovrapposte nella figura 13.




Figura 11: porzione di schiacciata lineare




Figura 12: porzione di schiacciata poligonale




Figura 13: porzione di schiacciata lineare e poligonale sovrapposte

Fase di riaccorpamento

dei poligoni:

in questa fase si analizza la situazione ottenuta intersecando il tema poligonale con la schiacciata lineare per ottenere la schiacciata poligonale. Occorre ricostruire i poligoni considerando però che dovranno avere, nei tratti dove deve essere garantita la congruenza, i contorni della schiacciata lineare di partenza. Perciò si procede a successive selezioni ed union utilizzando **l'apposito tool**.

Nelle figure 14-15 16-17 18-19 è possibile vedere degli esempi di selezione ed accorpamento




Figura 14: poligoni selezionati per il successivo accorpamento




Figura 15: Effetto dell'accorpamento




Figura 16: poligoni selezionati per il successivo accorpamento




Figura 17: Effetto dell'accorpamento




Figura 18: poligoni selezionati per il successivo accorpamento




Figura 19: Effetto dell'accorpamento

Notiamo l'ultimo poligono elaborato non può coincidere con i lati della schiacciata lineare, perché essendo archi appesi non hanno dato origine a poligoni nella schiacciata poligonale. Non è quindi possibile adottare in questo caso la tecnica d'accorpamento di poligoni. Si utilizza perciò il metodo di reshape come descritto dalle figure 20-21-22-23-24-25-26-27.




Figura 20: evidenziazione dei vertici della schiacciata




Figura 21: marcatura dell'arco di spezzata che fungerà da forma




Figura 22: marcatura del perimetro di poligono da rendere congruente




Figura 23: dopo il comando "esegui incolla spezzata" del poligono esterno




Figura 24: selezione del poligono interno




Figura 25: marcatura del perimetro di poligono interno da rendere congruente




Figura 26: dopo il comando “esegue incolla spezzata” del poligono interno




Figura 27: risultato